Defining the manifold:

As usual we need to import the xAct package , we will be needing both xTensor and xCoba for doing component computations.
In this notebook I have written down the steps for calculating the elements of Christoffel symbols, Riemann Curvature Tensor , etc from the metric of our choice.

In[1]:=
```
<< xAct`xTensor`
```

```
------------------------------------------------------------

Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}

CopyRight (C) 2003-2020, Jose M. Martin-Garcia, under the General Public License.

Connecting to external MinGW executable...

Connection established.

------------------------------------------------------------

Package xAct`xTensor`  version 1.1.5, {2021, 2, 28}

CopyRight (C) 2002-2021, Jose M. Martin-Garcia, under the General Public License.

------------------------------------------------------------

These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.

------------------------------------------------------------
```

Defining a manifold , all the objects we define in this notebook will be an defined in this 4 dimensional manifold.

In[2]:=
```
DefManifold[M4, 4, {α, β, γ, ν, λ, ρ, σ, η}]
```

```
** DefManifold: Defining manifold M4.

** DefVBundle: Defining vbundle TangentM4.
```

In[3]:=
```
DefMetric[-1, metric[-α, -β], CD, PrintAs → "g"]
```

```
** DefTensor: Defining symmetric metric tensor metric[-α, -β].
** DefTensor: Defining antisymmetric tensor epsilonmetric[-α, -β, -γ, -η].
** DefTensor: Defining tetrametric Tetrametric[-α, -β, -γ, -η].
** DefTensor: Defining tetrametric Tetrametric†[-α, -β, -γ, -η].
** DefCovD: Defining covariant derivative CD[-α].
** DefTensor: Defining vanishing torsion tensor TorsionCD[α, -β, -γ].
** DefTensor: Defining symmetric Christoffel tensor ChristoffelCD[α, -β, -γ].
** DefTensor: Defining Riemann tensor RiemannCD[-α, -β, -γ, -η].
** DefTensor: Defining symmetric Ricci tensor RicciCD[-α, -β].
** DefCovD:  Contractions of Riemann automatically replaced by Ricci.
** DefTensor: Defining Ricci scalar RicciScalarCD[].
** DefCovD:  Contractions of Ricci automatically replaced by RicciScalar.
** DefTensor: Defining symmetric Einstein tensor EinsteinCD[-α, -β].
** DefTensor: Defining Weyl tensor WeylCD[-α, -β, -γ, -η].
** DefTensor: Defining symmetric TFRicci tensor TFRicciCD[-α, -β].
** DefTensor: Defining Kretschmann scalar KretschmannCD[].
** DefCovD:  Computing RiemannToWeylRules for dim 4
** DefCovD:  Computing RicciToTFRicci for dim 4
** DefCovD:  Computing RicciToEinsteinRules for dim 4
** DefTensor: Defining weight +2 density Detmetric[]. Determinant.
```

Establishing the chart:

xCoba is required for component computations in xAct

In[4]:=
```
<< xAct`xCoba`
```

```
------------------------------------------------------------
Package xAct`xCoba`  version 0.8.6, {2021, 2, 28}
CopyRight (C) 2005-2021, David Yllanes
  and Jose M. Martin-Garcia, under the General Public License.
------------------------------------------------------------
These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.
------------------------------------------------------------
```

In[5]:=
```
$DefInfoQ = False;
$PrePrint = ScreenDollarIndices;
$CVSimplify = Simplify;
```

• Define a coordinate system:

schw is the name we assign for the Chart

the second value M4 is used to assign that this chart is define on the manifold M4
t,r,θ,ϕ are scalar fields whose respective numbering are numbered 0,1,2,3

In[8]:=
```
DefChart[schw, M4, {0, 1, 2, 3}, {t[], r[], θ[], ϕ[]}]
```

Next we need to define the constants that will be used in the metric, here the constants we use are C,G and mass (M)

In[9]:=
```
DefConstantSymbol[mass, PrintAs → "M"]
DefConstantSymbol[speedc, PrintAs → "C"]
DefConstantSymbol[G]
```

In[12]:=
```
a = 2 * G * mass / (speedc * speedc)
```

Out[12]=
$$\frac{2\,G\,M}{C^2}$$

Defining the metric elements:

Method 1:

here we are are assigning the elements of the metric, which are functions of the r[] and θ[] part of the chart schw we had defined earlier.
These are are components of the Schwarzschild metric and it is named as met

In[14]:=
```
MatrixForm[
 met = DiagonalMatrix[{-1 + a / r[], 1 / (1 - a / r[]), r[]^2, r[]^2 Sin[θ[]]^2}]]
```

Out[14]//MatrixForm=
$$\begin{pmatrix} -1+\frac{2\,G\,M}{C^2\,r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2\,G\,M}{C^2\,r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\mathrm{Sin}[\theta]^2 \end{pmatrix}$$

In[15]:=
```
MetricInBasis[metric, -schw, met] // TableForm
```

Added independent rule $g_{00} \to -1+\frac{2\,G\,M}{C^2\,r}$ for tensor metric

Added independent rule $g_{01} \to 0$ for tensor metric

Added independent rule $g_{02} \to 0$ for tensor metric

Added independent rule $g_{03} \to 0$ for tensor metric

Added dependent rule $g_{10} \to g_{01}$ for tensor metric

Added independent rule $g_{11} \to \frac{1}{1-\frac{2\,G\,M}{C^2\,r}}$ for tensor metric

Added independent rule $g_{12} \to 0$ for tensor metric

Added independent rule $g_{13} \to 0$ for tensor metric

Added dependent rule $g_{20} \to g_{02}$ for tensor metric

Added dependent rule $g_{21} \to g_{12}$ for tensor metric

Added independent rule $g_{22} \to r^2$ for tensor metric

Added independent rule $g_{23} \to 0$ for tensor metric

Added dependent rule $g_{30} \to g_{03}$ for tensor metric

Added dependent rule $g_{31} \to g_{13}$ for tensor metric

Added dependent rule $g_{32} \to g_{23}$ for tensor metric

Added independent rule $g_{33} \to r^2\,\mathrm{Sin}[\theta]^2$ for tensor metric

Out[15]//TableForm=

$g_{00} \to -1+\frac{2\,G\,M}{C^2\,r}$	$g_{01} \to 0$	$g_{02} \to 0$	$g_{03} \to 0$
$g_{10} \to 0$	$g_{11} \to \frac{1}{1-\frac{2\,G\,M}{C^2\,r}}$	$g_{12} \to 0$	$g_{13} \to 0$
$g_{20} \to 0$	$g_{21} \to 0$	$g_{22} \to r^2$	$g_{23} \to 0$
$g_{30} \to 0$	$g_{31} \to 0$	$g_{32} \to 0$	$g_{33} \to r^2\,\mathrm{Sin}[\theta]^2$

In[16]:=
```
MetricCompute[metric, schw, "Weyl"[-1, -1, -1, -1]]
```

Method 2:

here we are defining a CTensor object called g and setting it as the metric. the Signature of the metric {3,1,0} means there are 3 + and 1 - .

In[17]:=
```
g = CTensor[met, {-schw, -schw}];
```

In[31]:=
```
SetCMetric[g, -schw, SignatureOfMetric → {3, 1, 0}];
```

Here we can pick each component
choosing the element 00 ie tt of the Schwarzschild metric.

In[32]:=
```
g[{0, -schw}, {0, -schw}]
```

Out[32]= $-1 + \frac{2\,G\,M}{C^2\,r}$

In[33]:=
```
g[-α, -β]
```

Out[33]= $\boxed{\begin{pmatrix} -1+\frac{2GM}{C^2 r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2GM}{C^2 r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\mathrm{Sin}[\theta]^2 \end{pmatrix}}_{\alpha\beta}$

In[34]:=
```
MetricCompute[g, schw, "Weyl"[-1, -1, -1, -1]]
```

Defining the covariant derivative

In[35]:=
```
cd = CovDOfMetric[g]
```

Out[35]= CCovD[PDschw, CTensor[

$$\left\{\left\{\left\{0, \frac{G\,M}{r\,(-2\,G\,M + C^2\,r)}, 0, 0\right\}, \left\{\frac{G\,M}{r\,(-2\,G\,M + C^2\,r)}, 0, 0, 0\right\}, \{0, 0, 0, 0\}, \{0, 0, 0, 0\}\right\},\right.$$
$$\left\{\left\{-\frac{G\,M\,(2\,G\,M - C^2\,r)}{C^4\,r^3}, 0, 0, 0\right\}, \left\{0, \frac{G\,M}{2\,G\,M\,r - C^2\,r^2}, 0, 0\right\},\right.$$
$$\left.\left\{0, 0, \frac{2\,G\,M}{C^2} - r, 0\right\}, \left\{0, 0, 0, \frac{(2\,G\,M - C^2\,r)\,\mathrm{Sin}[\theta]^2}{C^2}\right\}\right\},$$
$$\left\{\{0, 0, 0, 0\}, \left\{0, 0, \frac{1}{r}, 0\right\}, \left\{0, \frac{1}{r}, 0, 0\right\}, \{0, 0, 0, -\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\}\right\},$$
$$\left.\left\{\{0, 0, 0, 0\}, \left\{0, 0, 0, \frac{1}{r}\right\}, \{0, 0, 0, \mathrm{Cot}[\theta]\}, \left\{0, \frac{1}{r}, \mathrm{Cot}[\theta], 0\right\}\right\}\right\},$$

$\{$schw, -schw, -schw$\}$, 0], CTensor$\left[\left\{\left\{-1 + \frac{2\,G\,M}{C^2\,r}, 0, 0, 0\right\}, \left\{0, \frac{1}{1-\frac{2\,G\,M}{C^2\,r}}, 0, 0\right\},\right.\right.$

$\left.\left.\{0, 0, r^2, 0\}, \{0, 0, 0, r^2\,\mathrm{Sin}[\theta]^2\}\right\}, \{-\mathrm{schw}, -\mathrm{schw}\}, 0\right]\Big]$

Christoffel Symbols in a coordinate Basis

In general we can write the Christoffel symbols as

In[36]:=
```
Christoffel[CD, PDschw][α, -β, -γ]
```

Out[36]= $\Gamma[\nabla,\mathcal{D}]^{\alpha}{}_{\beta\gamma}$

We can make a table of these in our coordinate basis.

In[37]:=
```
Part[TensorValues@ChristoffelCDPDschw, 2] // TableForm
```

Out[37]//TableForm=

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{00} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{01} \to \frac{G\,M}{r\,(-2\,G\,M+C^2\,r)}$

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{02} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{03} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{11} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{12} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{13} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{22} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{23} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{0}{}_{33} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{00} \to -\frac{G M \left(2 G M - C^2 r\right)}{C^4 r^3}$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{01} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{02} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{03} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{11} \to \frac{G M}{2 G M r - C^2 r^2}$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{12} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{13} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{22} \to \frac{2 G M}{C^2} - r$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{23} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{1}{}_{33} \to \frac{\left(2 G M - C^2 r\right) \mathrm{Sin}[\theta]^2}{C^2}$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{00} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{01} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{02} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{03} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{11} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{12} \to \frac{1}{r}$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{13} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{22} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{23} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{2}{}_{33} \to -\mathrm{Cos}[\theta]\, \mathrm{Sin}[\theta]$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{00} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{01} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{02} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{03} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{11} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{12} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{13} \to \frac{1}{r}$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{22} \to 0$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{23} \to \mathrm{Cot}[\theta]$

$\Gamma[\nabla,\mathcal{D}]^{3}{}_{33} \to 0$

Riemann Tensor

In[29]:=
```
riemann = Riemann[cd]
```

Out[29]=
```
CTensor[{{{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
  {{0, (2 G M (2 G M - C^2 r))/(C^4 r^4), 0, 0}, {-((2 G M)/(r^2 (-2 G M + C^2 r))), 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
  {{0, 0, -((G M (2 G M - C^2 r))/(C^4 r^4)), 0}, {0, 0, 0, 0}, {(G M)/(C^2 r), 0, 0, 0}, {0, 0, 0, 0}},
  {{0, 0, 0, -((G M (2 G M - C^2 r))/(C^4 r^4))}, {0, 0, 0, 0}, {0, 0, 0, 0}, {(G M Sin[θ]^2)/(C^2 r), 0, 0, 0}}},
 {{{0, -((2 G M (2 G M - C^2 r))/(C^4 r^4)), 0, 0}, {(2 G M)/(r^2 (-2 G M + C^2 r)), 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
  {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
  {{0, 0, 0, 0}, {0, 0, (G M)/(r^2 (2 G M - C^2 r)), 0}, {0, (G M)/(C^2 r), 0, 0}, {0, 0, 0, 0}},
  {{0, 0, 0, 0}, {0, 0, 0, (G M)/(r^2 (2 G M - C^2 r))}, {0, 0, 0, 0}, {0, (G M Sin[θ]^2)/(C^2 r), 0, 0}}},
 {{{0, 0, (G M (2 G M - C^2 r))/(C^4 r^4), 0}, {0, 0, 0, 0}, {-((G M)/(C^2 r)), 0, 0, 0}, {0, 0, 0, 0}},
  {{0, 0, 0, 0}, {0, 0, -((G M)/(r^2 (2 G M - C^2 r))), 0}, {0, -((G M)/(C^2 r)), 0, 0}, {0, 0, 0, 0}},
  {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}},
  {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, (2 G M)/(C^2 r)}, {0, 0, -((2 G M Sin[θ]^2)/(C^2 r)), 0}}},
 {{{0, 0, 0, (G M (2 G M - C^2 r))/(C^4 r^4)}, {0, 0, 0, 0}, {0, 0, 0, 0}, {-((G M Sin[θ]^2)/(C^2 r)), 0, 0, 0}},
  {{0, 0, 0, 0}, {0, 0, 0, -((G M)/(r^2 (2 G M - C^2 r)))}, {0, 0, 0, 0}, {0, -((G M Sin[θ]^2)/(C^2 r)), 0, 0}},
  {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, -((2 G M)/(C^2 r))}, {0, 0, (2 G M Sin[θ]^2)/(C^2 r), 0}},
  {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}}},
 {-schw, -schw, -schw, schw}, 0]
```

Printing elements using the indices:

In[26]:=
```
riemann[{0, -schw}, {1, -schw}, {0, -schw}, {1, schw}]
```

Out[26]=

$$\frac{2\,G\,M\left(2\,G\,M-C^2\,r\right)}{C^4\,r^4}$$

In[27]:=
```
riemann[{3, -schw}, {2, -schw}, {3, -schw}, {2, schw}]
```

Out[27]= $\frac{2\,G\,M\,\mathrm{Sin}[\theta]^2}{C^2\,r}$

The Ricci Tensor and Ricci Scalar:

In[38]:=
```
Ricci[cd][-α, -β]
```

Out[38]= 0

In[40]:=
```
rs = RicciScalar[cd]
```

Out[40]= Zero

Eienstien Tensor:

In[41]:=
```
Einstein[cd][-α, -β]
```

Out[41]= 0

Here we have used the Schwarzschild metric and computed the components of Christoffel symbols, Riemann Curvature tensor etc. The same procedure can be applied to any metric of our choice.
I will try to post more examples on component computations using xAct.
Feel Free to post any typos .